
PD Tema 13: Introducción a la programación lógica con Prolog

Lógica informática (2012–13)
Tema 13: Introducción a la programación lógica con Prolog

José A. Alonso Jiménez
Andrés Cordón Franco

María J. Hidalgo Doblado

Grupo de Lógica Computacional
Departamento de Ciencias de la Computación e I.A.

Universidad de Sevilla

1 / 43



PD Tema 13: Introducción a la programación lógica con Prolog

Tema 13: Introducción a la programación lógica con Prolog

1. El sistema deductivo de Prolog

2. Las listas en Prolog

3. Operadores en Prolog

4. Control mediante corte

5. Negación como fallo

2 / 43



PD Tema 13: Introducción a la programación lógica con Prolog
El sistema deductivo de Prolog

Tema 13: Introducción a la programación lógica con Prolog

1. El sistema deductivo de Prolog
Deducción Prolog en lógica proposicional
Deducción Prolog en lógica relacional
Deducción Prolog en lógica funcional

2. Las listas en Prolog

3. Operadores en Prolog

4. Control mediante corte

5. Negación como fallo

3 / 43



PD Tema 13: Introducción a la programación lógica con Prolog
El sistema deductivo de Prolog

Deducción Prolog en lógica proposicional

Deducción Prolog en lógica proposicional
I Base de conocimiento y objetivo:

I Base de conocimiento:
I Regla 1: Si un animal es ungulado y tiene rayas negras, entonces

es una cebra.
I Regla 2: Si un animal rumia y es mamífero, entonces es ungulado.
I Regla 3: Si un animal es mamífero y tiene pezuñas, entonces es

ungulado.
I Hecho 1: El animal es mamífero.
I Hecho 2: El animal tiene pezuñas.
I Hecho 3: El animal tiene rayas negras.

I Objetivo: Demostrar a partir de la base de conocimientos que el
animal es una cebra.

El sistema deductivo de Prolog

Deducción Prolog en lógica proposicional

Deducción Prolog en lógica proposicional
I Programa:

es_cebra :- es_ungulado, tiene_rayas_negras. %R1

es_ungulado :- rumia, es_mamífero. %R2

es_ungulado :- es_mamífero, tiene_pezuñas. %R3

es_mamífero. %H1

tiene_pezuñas. %H2

tiene_rayas_negras. %H3

I Sesión:
> pl

Welcome to SWI-Prolog (Multi-threaded, Version 5.6.20)

Copyright (c) 1990-2006 University of Amsterdam.

?- [animales].

Yes

?- es_cebra.

Yes

El sistema deductivo de Prolog

Deducción Prolog en lógica proposicional

Deducción Prolog en lógica proposicional
I Árbol de deducción:

El sistema deductivo de Prolog

Deducción Prolog en lógica proposicional

Deducción Prolog en lógica proposicional
I Demostración por resolución SLD:

El sistema deductivo de Prolog

Deducción Prolog en lógica relacional

Deducción Prolog en lógica relacional
I Base de conocimiento:

I Hechos 1-4: 6 y 12 son divisibles por 2 y por 3.
I Hecho 5: 4 es divisible por 2.
I Regla 1: Los números divisibles por 2 y por 3 son divisibles por 6.

I Programa:

divide(2,6). % Hecho 1

divide(2,4). % Hecho 2

divide(2,12). % Hecho 3

divide(3,6). % Hecho 4

divide(3,12). % Hecho 5

divide(6,X) :- divide(2,X), divide(3,X). % Regla 1

El sistema deductivo de Prolog

Deducción Prolog en lógica relacional

Deducción Prolog en lógica relacional
I Símbolos:

I Constantes: 2, 3, 4, 6, 12
I Relación binaria: divide
I Variable: X

I Interpretaciones de la Regla 1:
I divide(6,X) :- divide(2,X), divide(3,X).
I Interpretación declarativa:

(∀X )[divide(2,X ) ∧ divide(3,X ) → divide(6,X )]
I Interpretación procedimental.

I Consulta: ¿Cuáles son los múltiplos de 6?
?- divide(6,X).

X = 6 ;

X = 12 ;

No

El sistema deductivo de Prolog

Deducción Prolog en lógica relacional

Deducción Prolog en lógica relacional
I Árbol de deducción:

I Comentarios:
I Unificación.
I Cálculo de respuestas.
I Respuestas múltiples.

El sistema deductivo de Prolog

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Representación de los números naturales:

0, s(0), s(s(0)), . . .
I Definición de la suma:

0 + Y = Y

s(X) + Y = s(X+Y)

I Programa

suma(0,Y,Y). % R1

suma(s(X),Y,s(Z)) :- suma(X,Y,Z). % R2

I Consulta: ¿Cuál es la suma de s(0) y s(s(0))?
?- suma(s(0),s(s(0)),X).

X = s(s(s(0)))

Yes

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Árbol de deducción:

El sistema deductivo de Prolog

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Consulta:

I ¿Cuál es la resta de s(s(s(0))) y s(s(0))?
I Sesión:

?- suma(X,s(s(0)),s(s(s(0)))).

X = s(0) ;

No

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Árbol de deducción:

El sistema deductivo de Prolog

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Consulta:

I Pregunta: ¿Cuáles son las soluciones de la ecuación
X + Y = s(s(0))?

I Sesión:
?- suma(X,Y,s(s(0))).

X = 0 Y = s(s(0)) ;

X = s(0) Y = s(0) ;

X = s(s(0)) Y = 0 ;

No

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Árbol de deducción:

El sistema deductivo de Prolog

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Consulta:

I Pregunta: resolver el sistema de ecuaciones
1+ X = Y
X + Y = 1

I Sesión:
?- suma(s(0),X,Y), suma(X,Y,s(0)).

X = 0

Y = s(0) ;

No

El sistema deductivo de Prolog

Deducción Prolog en lógica funcional

Deducción Prolog en lógica funcional
I Árbol de deducción:

Las listas en Prolog

Representación de listas

?- [X|Y] = [a,b].

X = a

Y = [b]

?- [X|Y] = [a,b,c,d].

X = a

Y = [b, c, d]

?- [X,Y|Z] = [a,b,c,d].

X = a

Y = b

Z = [c, d]

Las listas en Prolog

Definición de relaciones sobre listas

Definición de concatenación (append)
I Especificación: conc(A,B,C) se verifica si C es la lista obtenida

escribiendo los elementos de la lista B a continuación de los
elementos de la lista A. Por ejemplo,
?- conc([a,b],[b,d],C).

C =[a,b,b,d]

I Definición 1:

conc(A,B,C) :- A=[], C=B.

conc(A,B,C) :- A=[X|D], conc(D,B,E), C=[X|E].

I Definición 2:

conc([],B,B).

conc([X|D],B,[X|E]) :- conc(D,B,E).

Las listas en Prolog

Definición de relaciones sobre listas

Consultas con la relación de concatenación
I Analogía entre la definición de conc y la de suma,
I ¿Cuál es el resultado de concatenar las listas [a,b] y [c,d,e]?

?- conc([a,b],[c,d,e],L).

L = [a, b, c, d, e]
I ¿Qué lista hay que añadirle a la lista [a,b] para obtener

[a,b,c,d]?
?- conc([a,b],L,[a,b,c,d]).

L = [c, d]
I ¿Qué dos listas hay que concatenar para obtener [a,b]?

?- conc(L,M,[a,b]).

L = [] M = [a, b] ;

L = [a] M = [b] ;

L = [a, b] M = [] ;

No

22 / 43



PD Tema 13: Introducción a la programación lógica con Prolog
Las listas en Prolog

Definición de relaciones sobre listas

Árbol de deducción de ?- conc(L,M,[a,b]).

Las listas en Prolog

Definición de relaciones sobre listas

Definición de la relación de pertenencia (member)
I Especificación: pertenece(X,L) se verifica si X es un elemento

de la lista L.
I Definición 1:

pertenece(X,[X|L]).

pertenece(X,[Y|L]) :- pertenece(X,L).

I Definición 2:

pertenece(X,[X|_]).

pertenece(X,[_|L]) :- pertenece(X,L).

24 / 43



PD Tema 13: Introducción a la programación lógica con Prolog
Las listas en Prolog

Definición de relaciones sobre listas

Consultas con la relación de pertenencia
?- pertenece(b,[a,b,c]).

Yes

?- pertenece(d,[a,b,c]).

No

?- pertenece(X,[a,b,a]).

X = a ;

X = b ;

X = a ;

No

?- pertenece(a,L).

L = [a|_G233] ;

L = [_G232, a|_G236] ;

L = [_G232, _G235, a|_G239]

Yes

Las listas en Prolog

Definición de relaciones sobre listas

Árbol de deducción de ?- pertenece(a,L).

Operadores en Prolog

Ejemplos de operadores aritméticos
I Ejemplos de notación infija y prefija en expresiones aritméticas:

?- +(X,Y) = a+b.

X = a

Y = b

?- +(X,Y) = a+b+c.

X = a+b

Y = c

?- +(X,Y) = a+(b+c).

X = a

Y = b+c

?- a+b+c = (a+b)+c.

Yes

?- a+b+c = a+(b+c).

No

Operadores en Prolog

Ejemplos de asociatividad y precedencia
I Ejemplos de asociatividad:

?- X^Y = a^b^c.

X = a Y = b^c

?- a^b^c = a^(b^c).

Yes

I Ejemplo de precedencia
?- X+Y = a+b*c.

X = a Y = b*c

?- X*Y = a+b*c.

No

?- X*Y = (a+b)*c.

X = a+b Y = c

?- a+b*c = (a+b)*c.

Operadores en Prolog

Definición de operadores
I Definición (ejemplo_operadores.pl)

:-op(800,xfx,estudian).

:-op(400,xfx,y).

juan y ana estudian lógica.

I Consultas
?- [ejemplo_operadores].

?- Quienes estudian lógica.

Quienes = juan y ana

?- juan y Otro estudian Algo.

Otro = ana

Algo = lógica

Control mediante corte

Ejemplo de nota sin corte
I nota(X,Y) se verifica si Y es la calificación correspondiente a la

nota X; es decir, Y es suspenso si X es menor que 5, Y es
aprobado si X es mayor o igual que 5 pero menor que 7, Y es
notable si X es mayor que 7 pero menor que 9 e Y es
sobresaliente si X es mayor que 9. Por ejemplo,
?- nota(6,Y).

Y = aprobado;

No

nota(X,suspenso) :- X < 5.

nota(X,aprobado) :- X >= 5, X < 7.

nota(X,notable) :- X >= 7, X < 9.

nota(X,sobresaliente) :- X >= 9.

32 / 43



PD Tema 13: Introducción a la programación lógica con Prolog
Control mediante corte

Deducción en el ejemplo sin corte
I Árbol de deducción de ?- nota(6,Y).

Control mediante corte

Ejemplo de nota con cortes
I Definición de nota con cortes

nota(X,suspenso) :- X < 5, !.

nota(X,aprobado) :- X < 7, !.

nota(X,notable) :- X < 9, !.

nota(X,sobresaliente).

Control mediante corte

Deducción en el ejemplo con cortes

I ¿Un 6 es un sobresaliente?
?- nota(6,sobresaliente).

Yes

Negación como fallo

Definición de la negación como fallo
I Definición de la negación como fallo not:

no(P) :- P, !, fail. % No 1

no(P). % No 2

Negación como fallo

Programa con negación
I Programa:

aprobado(X) :- % R1

no(suspenso(X)),

matriculado(X).

matriculado(juan). % R2

matriculado(luis). % R3

suspenso(juan). % R4

I Consultas:
?- aprobado(luis).

Yes

?- aprobado(X).

No

Negación como fallo

Árbol de deducción de ?- aprobado(luis).

Negación como fallo

Árbol de deducción de ?- aprobado(X).

Negación como fallo

Modificación del orden de los literales
I Programa:

aprobado(X) :- % R1

matriculado(X),

no(suspenso(X)).

matriculado(juan). % R2

matriculado(luis). % R3

suspenso(juan). % R4

I Consulta:
?- aprobado(X).

X = luis

Yes

Árbol de deducción de ?- aprobado(X).

